80-781

сообщения
объединенного
института
ядерных
исследований
дубна

P11-80-781

В.Гаджоков, Н.Богданова

ПРОГРАММЫ АППРОКСИМАЦИИ МЕТОДОМ НАИМЕНЬШИХ КВАДРАТОВ С ИСПОЛЬЗОВАНИЕМ ОРТОНОРМИРОВАННЫХ ПОЛИНОМОВ

1980

1. ВВЕДЕНИЕ

В предыдущих работах [1,2] мы описали алгоритм построения набора полиномов $\{P_j\}$, ортонормированных на произвольном конечном дискретном действительном множестве, а также способ использования этих полиномов для аппроксимации измеренных на опыте зависимостей. Цель настоящей работы - представить программы, реализующие изложенный ранее метод в виде, готовом для применения. Несмотря на давнюю известность этого метода [3,4] и на неоднократные его программные реализации /см. [5-8] - правда, все в ненормализованном виде/, нам кажется, что он используется значительно реже, чем можно было бы ожидать. Напомним, что речь идет о построении такого набора полиномов, который превращает матрицу системы нормальных уравнений в единичную и тем самым делает ненужным ее обращение. В этом отношении рассматриваемый метод отличается как от подхода [9], где из-за ортогонализации на непрерывной области та же матрица получается приблизительно диагональной и поэтому приходится ее инвертировать, так и от [10], где применяется прямая диагонализация матриц без нормализации.

2. ФОРМУЛИРОВКА ЗАДАЧИ

Пусть множества $\{x_i\}$, $\{y_i\}$ и $\{\Delta y_i\}$, $i = 1, 2, \ldots, M$ заданы. Здесь x_i - значения независимой переменной, y_i - значения аппроксимируемой функции и Δy_i - стандартные отклонения y_i, через которые выражаются веса w_i заданных точек

$$w_i = 1/(\Delta y_i)^2. \qquad /3.1/$$

Требуется найти коэффициенты a_j, $j = 0, 2, \ldots, k$ разложения

$$y(x) = \sum_{j=0}^{k} a_j P_j(x), \qquad /3.2/$$

которые удовлетворяли бы условию

$$\chi^2 = \sum_{i=1}^{M} w_i (y_i - y(x_i))^2 \to \min, \qquad /3.3/$$

а также оценить наследственные ошибки коэффициентов Δa_j. Наивысший порядок k полинома в ряде /3.2/ также заранее неизвестен.

3. МЕТОД РЕШЕНИЯ

При решении поставленной задачи мы будем придерживаться методики [2] со следующими дополнительными замечаниями.

3.1. Как во время вычисления коэффициентов ортонормированного ряда и их наследственных ошибок, так и при использовании полученного ряда, необходим многократный расчет сумм типа /3.2/. В принципе это возможно при помощи обращения к программе ØRTHØN [1], которая выдает значения P_0, $P_1(x)$, $P_2(x)$,..., $P_k(x)$ при любом заданном аргументе x. Доказано, однако, что суммы этого типа вычисляются быстрее и точнее [11] на основе рекуррентного соотношения, связывающего полиномы $P_{i-1}(x)$, $P_i(x)$ и $P_{i+1}(x)$, что приводит к выражению

$$\sum_{j=0}^{k} a_j P_j(x) = K_0/\beta_0, \qquad /3.4/$$

где

$$K_k = a_k, \qquad /3.5/$$

$$K_{k-1} = a_{k-1} + \frac{x-\alpha_k}{\beta_k} a_k, \qquad /3.6/$$

$$K_i = a_i + \frac{x-\alpha_{i+1}}{\beta_{i+1}} - \frac{\beta_{i+1}}{\beta_{i+2}} K_{i+2},$$

$$i = k-2, k-3, \ldots, 1, 0, \qquad /3.7/$$

а α_i и β_i - рекуррентные коэффициенты из [1].

В описываемых программах расчет сумм типа /3.2/ проводится при помощи формул /3.4/-/3.7/, реализованных в подпрограмме-функции TELSC1.

3.2. Для удобства пользователя, привыкшего работать с коэффициентами обычного полиномиального ряда, мы включили в пакет программу CØNV1, которая вычисляет коэффициенты c_j разложения

$$\sum_{j=0}^{k} a_j P_j(x) = \sum_{j=0}^{k} c_j (x')^j, \qquad /3.8/$$

где

$$x' = x_1 x + x_0, \qquad /3.9/$$

$$x_1 = \frac{2}{x_{max} - x_{min}}, \qquad /3.10/$$

$$x_0 = -\frac{x_{max} + x_{min}}{x_{max} - x_{min}}. \qquad /3.11/$$

Нетрудно видеть, что линейное преобразование /3.9/ переводит рабочий интервал значений $x \in [x_{min}, x_{max}]$ в отрезок $[-1, +1]$. Коэффициенты c_j вычисляются по формулам, вытекающим из трехчленного рекуррентного соотношения

$$c_j = \sum_{i=j}^{k} a_i a_j^{(i)}, \qquad /3.12/$$

где $a_j^{(i)}$ - коэффициенты при x^j в полиноме $P_i(x)$, причем

$$a_0^{(0)} = 1/\beta_0, \qquad /3.13/$$

и имеет место рекурсия

$$a_j^{(i+1)} = \frac{1}{\beta_{i+1}} [(1-\delta_{0j}) a_{j-1}^{(i)} - (1-\delta_{i+1,j}) \alpha_{i+1} a_j^{(i)} - (1-\delta_{ij})(1-\delta_{i+1,j}) \beta_i a_j^{(i-1)}], \qquad /3.14/$$

где δ_{ij} - символ Кронекера.

Коэффициенты c_j и их наследственные ошибки могут оказаться полезными при оценке качества аппроксимации. Что касается расчета функции $y(x)$, то мы отдаем предпочтение подходу, описанному в предыдущем разделе.

3.3. Существует несколько способов подбора оптимальной длины k аппроксимирующего ряда /3.2/ - /см., напр., /4,12/ /.

Исходя из требования гладкости аппроксимирующей функции, из статистических соображений и из опыта обработки экспериментальных данных, мы остановились на двух из них:

а/ минимальном k, при котором неравенство

$$w_i [y_i - y(x_i)]^2 \leq 1 \qquad /3.15/$$

выполняется для всех $i = 1, 2, \ldots, M$;

б/ значении k, при котором величина χ^2, нормализованная при числе степеней свободы $M-k-1$, достигает абсолютного минимума.

Отметим, что условие а требует, чтобы аппроксимирующая кривая проходила внутри коридоров ошибок всех заданных точек. Если оно удовлетворено, то поиск более высокого k, удовлетворяющего условию b, прекращается; в этом смысле первое условие сильнее, т.е. ему отдается предпочтение.

4. ОПИСАНИЕ ПРОГРАММ

Программная реализация метода основывается на пакете из трех SUBRØUTINES, опубликованных ранее /1/, со следующими изменениями:

а/ В целях унификации названий программы переименованы:

PØPTHN	в	PRERF1,
ØRTHØN	в	ØRTHN1,
ERØRTH	в	ERR1.

б/ Так как оказалось, что условие упорядоченности множества $\{x_i\}$ часто является помехой непосредственному применению программ, то это условие снято заменой первых четырех выполняемых операторов программы PRERF1 (PØRTHN) на:

```
XMIN = X(1)
XMAX = X(1)
DØ5 I=2,M
```

```
IF(X(I).LT.XMIN)    XMIN = X(I)

IF(X(I).GT.XMAX)  XMAX = X(I)

5 CØNTINUE.
```

Таким образом, PRERF1 сама находит крайние значения аргумента x_{min} и x_{max}, и предварительная сортировка массивов по величине x_i становится ненужной.

Передача значений входных данных и результатов вычислений осуществляется через именованный CØMMØN|LINKS|, в который входят в указанном порядке:

X(100) - массив значений аргумента $\{x_i\}$, $i=1,2,...,M$;
Y(100) - массив значений функции $\{y_i\}$;
W(100) - массив значений весов $\{w_i\}$;
PA(100) - массив значений аппроксимированной функции $\{y(x_i)\}$;
DELTA(100) - массив значений разностей $\{y_i - y(x_i)\}$;
A(26) - коэффициенты $\{a_j\}$ ряда /3.2/;
ERRA(26) - оценки ошибок $\{\Delta a_j\}$;
ALPHA(25) - рекуррентные коэффициенты /см. /1//;
BETA(26) - нормирующие коэффициенты /см. /1/ /;
PØLY(27) - рабочий массив значений полиномов $\{P_j\}$.

Именованный CØMMON|LINKS| должен быть заявлен в вызывающей программе, а исходные данные занесены в X, Y, W до обращения к программе аппроксимации.

4.1. Программа APPRØ1 осуществляет собственно аппроксимацию, т.е. вычисляет a_j и проводит поиск оптимальной длины ортонормированного ряда. Она вызывается командой CALL APPRØ1 (M, LMIN, LMAX, LØPT, THØPT), где M - число входных точек, LMIN и LMAX определяют замкнутый отрезок, на котором ищется оптимальное значение k=LØPT, а в THOPT возвращается величина

$$THOPT = \left(\frac{\chi^2}{M-k-1}\right)^{1/2}, \qquad /3.16/$$

соответствующая k = LØ PT.

4.2. Программа ØUT1 проводит расчет наследственных ошибок Δa_j и осуществляет вывод результатов в виде стандартной таблицы, чей смысл ясен из заголовков колонок. Ее вызов:

CALL : ØUT1 (M, LØPT, THØPT).

4.3. Программа CØNV1 преобразует ряд /3.2/ в вид /3.8/ и выдает на печать значения $c_j \pm \Delta c_j$. Она должна вызываться после

APPRØ1 и ØUT1 командой CALL CØNV1(A, ERRA, ALPHA, BETA, C, ERRC, LØPT). Если этот вызов необходим, то в главной программе должна резервироваться память под массивы C и ERRC, напр., DIMENSION C(26), ERRC (26).

4.4. Программа TELSC1 суммирует ряд /3.2/ по методу "телескопирования" /11/ /см. формулы /3.4/-/3.7//. Она может быть полезной при расчете функции y(x) в точке, не принадлежащей исходному множеству. Эта программа используется в APPRØ1 и вызывается обращением к ней с аргументами A, ALPHA, BETA, XX, LOPT. Отметим, что XX тут не массив, а простая переменная, при значении которой TELSC1 вычисляет ряд /3.2/, совершив заранее преобразование /3.9/.

4.5. Программа HØRN1 вычисляет ряд правой части /3.8/ по методу Горнера, используя при этом коэффициенты, полученные на выходе CØNV1. Она вызывается также обращением к CØNV1 с аргументами C,K,X, где C - массив значений c_j, K - наивысшая степень независимой переменной x, X - значение независимой переменной. Эта величина должна быть задана в своих естественных единицах, т.к. преобразование /3.9/ осуществляется самой программой HØRN1. Естественно, выход программ TELSC1 и HORN1 совпадает в пределах точности представления чисел в используемой ЭВМ.

4.6. Тексты пяти программ на фортране-IV воспроизведены в приложении. Там же дана и главная программа TCONV1, которая иллюстрирует использование пакета для аппроксимации функции $y=x^3$ на том же /пятиточечном/ примере в /1/.

5. ЗАМЕЧАНИЯ ОБ ИСПОЛЬЗОВАНИИ ПРОГРАММ

Пакет программ был запущен на машинах ИЗОТ-0310, ЕС-1010, ЕС-1040 и CDC-6500. Решение тестовых задач на этих ЭВМ приводит к одинаковым в пределах машинной точности результатам. В описываемом варианте допускается максимальное число точек /100/ и максимальная степень полиномов /25/. Эти ограничения можно изменить путем соответствующих модификаций именованного COMMOM|LINKS|. Кроме задач, описанных в /2/, этим методом был нами аппроксимирован ряд других зависимостей, в частности, зависимость коэффициентов внутренней конверсии от энергии γ-излучения, кулоновского формфактора β-спектра от энергии, а также от координат точек реперных крестов при бесфильмовом съеме информации со стримерной камеры установки РИСК. Во всех этих случаях мы получили удовлетворительные результаты при точности мантисс в 23 бита, причем время обработки было значительно ниже, чем при традиционных методах. Представляется интересной возможность использования этой методики для аппроксимации и интерполяции функций, чьи значения известны с ограниченной точностью, а методы прямого расчета требуют значительных затрат машинного времени - напр., при тех же коэффициентах внутренней конверсии. В этих условиях описываемый подход приводит к резкой свертке хранимой информации /т.е. коэффициентов a_j /. По этому показателю он превосходит метод сплайнов, а неточность входных данных освобождает нас от необходимости проводить аппроксимацию, которая восстанавливала бы точные значения функции в точках сетки. Отсутствие жестких условий на расположение точек сетки и на величины весов облегчает применение программ в случае неравномерного табличного шага и разных точностей определения величин $\{y_i\}$.

В заключение отметим, что восьмиточечный пример, чье решение обсуждается в нескольких вариантах на стр. 218 монографии /13/, решается нашими программами на ЭВМ ИЗОТ-310 за несколько секунд, а результат совпадает с наилучшим вариантом, приведенным в книге. При этом мы использовали единичную точность и, конечно, нам не приходилось обращаться к сингулярному анализу.

Авторы выражают благодарность Б.Н.Покровскому за интерес к работе и за содействие в поиске более ранних публикаций по построению систем ортогональных полиномов.

ПРИЛОЖЕНИЕ

Тексты программ на ФОРТРАНе-IV

```
C TEST EXTENDED ONE-DIMENSIONAL ORTHONORMAL PACKAGE
      COMMON /LINKS/ X(100),W(100),Y(100),PA(100),DELTA(100),
     1               A(26),ERRA(26),ALPHA(25),BETA(26),POLY(27)
      DIMENSION C(26),ERRC(26)
      DO 5 I=1,5
      X(I)=.5*FLOAT(I)-1.5
      Y(I)=X(I)**3
    5 W(I)=5000.
      W(3)=20000.
      CALL APPRO1(5,1,4,LOPT,THOPT)
      CALL OUT1(5,LOPT,THOPT)
      CALL CONV1(A,ERRA,ALPHA,BETA,C,ERRC,LOPT)
      DO 10 I=1,5
      YTEL=TELSC1(A,ALPHA,BETA,X(I),LOPT)
      YHOR=HORN1(C,LOPT,X(I))
      DTEL=Y(I)-YTEL
      DHOR=Y(I)-YHOR
      DMUT=YTEL-YHOR
   10 WRITE(1,20)X(I),Y(I),YTEL,YHOR,DTEL,DHOR,DMUT
   20 FORMAT(0P7E15.5)
      STOP
      END
```

```
      SUBROUTINE OUT1(M,L,THETA)
      COMMON /LINKS/ X(100),W(100),Y(100),PA(100),DELTA(100),
     1               A(26),ERRA(26),ALPHA(25),BETA(26),POLY(27)
      WRITE(1,200)L
  200 FORMAT('1ORTHONORMAL POLYNOMIAL APPROXIMATION'/
     1       ' ***********************************'/
     2       58X,'L=',I2/
     3       11X,'X',15X,'W',13X,'YEXP',12X,'YAPPR',8X,'YEXP-YAPPR'/
     4       41H ======================================== ,
     5       40H========================================  )
C RESTORE OPTIMUM VALUES OF PA AND DELTA ARRAYS
      DO 170 J=1,M
      PA(J)=TELSC1(A,ALPHA,BETA,X(J),L)
      DELTA(J)=Y(J)-PA(J)
      WRITE(1,166)X(J),W(J),Y(J),PA(J),DELTA(J)
  166 FORMAT(0P5E16.4)
  170 CONTINUE
      LLL=L+1
      WRITE(1,175)THETA
  175 FORMAT('   NORMALIZED CHI-SQUARE:',0PE16.5)
C
C ERROR COMPUTATION ROUTINE
C
C PRINT HEADER
      WRITE(1,1001)
 1001 FORMAT(1H0,3X,'I',10X,'A(I)',13X,'ERROR')
C BEGIN EXTERNAL CYCLE              (INDEX IEXT)
      DO 1800 IEXT=1,M
      CALL ORTHN1(L,X(IEXT),ALPHA,BETA,POLY)
C BEGIN INTERNAL CYCLE              (INDEX I1)
          DO 1700 I1=1,LLL
 1700     ERRA(I1)=ERRA(I1)+(POLY(I1+1)*W(IEXT)*DELTA(IEXT))**2
 1800 CONTINUE
C BOTH LOOPS COMPLETED
      DO 1900 I=1,LLL
      IPEXT=I-1
      ERRA(I)=SQRT(ERRA(I))
 1900 WRITE(1,1930)IPEXT,A(I),ERRA(I)
 1930 FORMAT(I5,0P2E18.5)
      RETURN
      END


      SUBROUTINE APPRO1(M,LMIN,LMAX,LOPT,THETAC)
C FITS Y(X) BY MEANS OF AN ORTHONORMAL SERIES
C SEEKS FOR OPTIMUM SERIES WITHIN DEGREE RANGE [LMIN,LMAX]
      COMMON /LINKS/ X(100),W(100),Y(100),PA(100),DELTA(100),
     1               A(26),ERRA(26),ALPHA(25),BETA(26),POLY(27)
C PREPARE RECURRENCE FACTORS
      CALL PRERF1(M,LMAX,X,W,ALPHA,BETA,POLY)
C COMPUTE COEFFICIENTS OF FITTING SERIES
      LLL=LMAX+1
      DO 125 I=1,LLL
      A(I)=0.0
  125 ERRA(I)=0.0
        DO 160 K=1,M
        CALL ORTHN1(LMAX,X(K),ALPHA,BETA,POLY)
           DO 160 I=1,LLL
  160      A(I)=A(I)+POLY(I+1)*W(K)*Y(K)
```

```
C APPROXIMATION COMPLETED; BEGIN SELECTING THE OPTIMUM
      THETAC=1.0E+19
      DO 3000 L=LMIN,LMAX
      IFLAG=0
      THETA=0.0
         DO 170 J=1,M
         PA(J)=TELSC1(A,ALPHA,BETA,X(J),L)
         DELTA(J)=Y(J)-PA(J)
         D2W=W(J)*DELTA(J)**2
         IF(IFLAG.NE.0) GO TO 170
         IF(D2W.GT.1.) IFLAG=1
  170    THETA=THETA+D2W
      IDF=M-L-1
      IF(IDF.EQ.0) IDF=1
      THETA=SQRT(THETA/FLOAT(IDF))
      IF(IFLAG.NE.1) GO TO 3010
      IF(THETA.GE.THETAC) GO TO 3000
      LOPT=L
      THETAC=THETA
 3000 CONTINUE
      GO TO 3020
 3010 LOPT=L
      THETAC=THETA
 3020 RETURN
      END


      SUBROUTINE CONV1(A,ERRA,ALPHA,BETA,C,ERRC,K)
C CONVERSION FROM ORTHONORMAL TO ORDINARY POWER SERIES
C     A     - COEFFICIENTS OF ORTHONORMAL SERIES
C     ERRA  - STANDARD ERRORS OF A
C     ALPHA - RECURRENCE FACTORS OF ORTHONORMAL SET
C     BETA  - NORMALIZING FACTORS OF ORTHONORMAL SET
C     C     - COEFFICIENTS OF ORDINARY POWER SERIES
C     ERRC  - STANDARD ERRORS OF C
C     K     - MAXIMUM DEGREE OF BOTH SERIES
      DIMENSION A(K),ERRA(K),ALPHA(K),BETA(K),C(K),ERRC(K)
      DIMENSION AA(26),AM1(26),AM2(26)
C NOTE: K SHOULD NOT EXCEDE 25
C CLEAR ARRAYS, SET INITIAL VALUE OF AM1(1)
      KK=K+1
         DO 10 I=1,KK
         C(I)=0.0
   10    ERRC(I)=0.0
      AM1(1)=1.0/BETA(1)
      C(1)=A(1)*AM1(1)
C BEGIN EXTERNAL LOOP (ORDER OF POLYNOMIALS)
      DO 300 II=1,K
      I=II-1
      ILIM=II+1
C BEGIN FIRST INTERMEDIATE LOOP (DEGREE OF VARIABLE)
            DO 200 JJ=1,ILIM
            J=JJ-1
C APPLY RECURRENCY TO COEFFICIENTS AA(J) FOR J=0,I+1
            AA(JJ)=0.0
            IF(J.NE.0)                   AA(JJ)=AA(JJ)+AM1(J)
            IF(J.NE.II)                  AA(JJ)=AA(JJ)-ALPHA(II)*AM1(JJ)
            IF((J.NE.II).AND.(J.NE.I))   AA(JJ)=AA(JJ)-BETA(II)*AM2(JJ)
  200       AA(JJ)=AA(JJ)/BETA(ILIM)
```

```
C FIRST INTERMEDIATE LOOP COMPLETED, BEGIN THE SECOND
                    DO 210 JJ=1,ILIM
                    ERRC(JJ)=ERRC(JJ)+(ERRA(ILIM)*AA(JJ))**2
  210               C(JJ)=C(JJ)+A(ILIM)*AA(JJ)
C SECOND INTERMEDIATE LOOP (SUM UP) COMPLETED, BEGIN THE THIRD
                          DO 220 J=1,ILIM
                          AM2(J)=AM1(J)
  220                     AM1(J)=AA(J)
C THIRD INTERMEDIATE LOOP COMPLETED (RECURRENCY SHIFT)
  300 CONTINUE
C ALL LOOPS COMPLETED
C CONTROL PRINT
      WRITE(1,401)
      DO 400 I=1,KK
      II=I-1
      ERRC(I)=SQRT(ERRC(I))
  400 WRITE(1,402)II,C(I),ERRC(I)
  401 FORMAT(//' NEW COEFFICIENTS:'/'  I',7X,'C(I)',8X,'ERROR')
  402 FORMAT(I3,0P2E13.5)
      RETURN
      END


      FUNCTION TELSC1(A,ALPHA,BETA,X,N)
C COMPUTE THE SUM A(I)*POLYI(X) OVER I=1,2,3,...,N
C BY MEANS OF THE TELESCOPING TRICK;  0 <= N <= NPREP
      COMMON /ORTNRM/X0,X1,MESSG,NPREP,XX
      DIMENSION A(N),ALPHA(N),BETA(N)
      REAL K,K1,K2
C LINEAR TRANSFORM OF ARGUMENT
      XX=X0+X1*X
C INITIALIZATION AND BRANCH ON N
      K2=A(N+1)
      K=K2
      IF (N.EQ.0) GO TO 20
      K1=A(N)+K2*(XX-ALPHA(N))/BETA(N+1)
      K=K1
      IF (N.EQ.1) GO TO 20
C BEGIN RECURSION LOOP - ROTATES FROM J=N-2 TO 0 (N-1 TIMES BACKWORDS)
        DO 10 I=2,N
        J=N-I
        K=A(J+1)+K1*(XX-ALPHA(J+1))/BETA(J+2)-K2*BETA(J+2)/BETA(J+3)
        K2=K1
   10   K1=K
   20 TELSC1=K/BETA(1)
      RETURN
      END


      FUNCTION HORN1(C,K,X)
      COMMON /ORTNRM/ X0,X1,MESSG,NPREP,XX
      DIMENSION C(K)
      XX=X1*X+X0
      J=K+1
      HORN1=C(J)
      DO 10 I=1,K
      J=J-1
   10 HORN1=HORN1*XX+C(J)
      RETURN
      END
```

ЛИТЕРАТУРА

1. Гаджоков В., Богданова Н. ОИЯИ, Р11-12860, Дубна, 1979.
2. Гаджоков В., Богданова Н. ОИЯИ, Р11-80-122, Дубна, 1980.
3. Forsythe G.E., J.Soc. Industr. Appl. Math., 1957, 5, pp. 74-88.
4. Худсон Д. Статистика для физиков. "Мир", М., 1967.
5. Ascher M., Forsythe G.E. J. Assoc.Comp., 1958, 5, No. 1, pp. 9-21.
6. Cooper B.E., Roy J. Stat.Soc.Ser. C (Appl.Stat.) 1968, 17, No. 3, p. 283.
7. Mackinney J.G. Comm. ACM(Nov 1960), 3, p. 604.
8. Makinson J.J. Comm. ACM, (Feb. 1967), 10, No.2, pp. 87-88.
9. Байла И., Ососков Г.А., Хэрн А.К. ОИЯИ, Р10-11944, Дубна, 1978.
10. Байла И., Ососков Г.А. ОИЯИ, Р10-11834, Дубна, 1978.
11. Beckmann P. "Orthogonal Polynomials for Engineers and Physicsts", The Golem Press, Boulder (Colorado, USA),1973.
12. Клепиков Н.Н., Соколов С.Н. Анализ и планирование экспериментов методом максимума правдоподобия , "Наука", М., 1964.
13. Форсайт Дж., Малькольм М., Моулер К. Машинные методы математических вычислений , "Мир", М., 1980.

Рукопись поступила в издательский отдел
1 декабря 1980 года.